

Chapter 2

Extensions of the Simple Model

- M21.GMS two goods, two factors, one household (same as M1_MPS.GMS)
- M22.GMS introduces intermediate inputs and nesting
- M23.GMS introduces joint production
- M24.GMS introduces the use of specific factors
- M25.GMS use of an initially stack activity (e.g., modeling tax avoidance)
- M26.GMS introduces a labor supply or labor/leisure choice activity
- M27.GMS two forms of labor supply, such as to formal/informal sectors
- M28.GMS two households with different preferences and endowments
- M29.GMS introduces Stone-Geary (LES) preferences, non-homothetic demand

Model M22 introduces intermediate inputs and nesting

This model is shown below. X and Y sectors each use the other sector's output as an input, so that each sector has three inputs.

With three inputs, there may be different elasticities of substitution between different pairs of these inputs.

```
$PROD:Y  s:0.75  va:1
          O:PY    Q:120
          I:PX    Q: 20
          I:PL    Q: 60  va:
          I:PK    Q: 40  va:
```

This notation specifies that labor and capital are in a “nest”, with the elasticity of substitution between capital and labor given by $va = 1$ (Cobb-Douglas).

We use va for “value added”, you can use anything you want. The technology also specifies that the elasticity of substitution between value added and $X = 0.75$.

\$TITLE Model M22: Closed Economy 2X2 with Intermediate
Inputs and Nesting

\$ONTEXT

Markets	Production Sectors			Consumers
	X	Y	W	CONS
PX	120	-20	-100	
PY	-20	120	-100	
PW			200	-200
PL	-40	-60		100
PK	-60	-40		100

\$OFFTEXT

PARAMETERS

TX;

TX = 0;

\$ONTEXT

\$MODEL: M22

\$SECTORS:

X ! Activity level for sector X

Y ! Activity level for sector Y

W ! Activity level for sector W (Hicksian

welfare index)

\$COMMODITIES:

PX ! Price index for commodity X

PY ! Price index for commodity Y

PL ! Price index for primary factor L

PK ! Price index for primary factor K

PW ! Price index for welfare

\$CONSUMERS:

CONS ! Income level for consumer CONS

\$PROD:X s:0.5 va:1
 O:PX Q:120
 I:PY Q: 20
 I:PL Q:40 va: A:CONS T:TX
 I:PK Q:60 va: A:CONS T:TX

\$PROD:Y s:0.75 va:1
 O:PY Q:120
 I:PX Q: 20
 I:PL Q: 60 va:
 I:PK Q: 40 va:

\$PROD:W s:1
 O:PW Q:200
 I:PX Q:100
 I:PY Q:100

\$DEMAND:CONS
 D:PW Q:200
 E:PL Q:100
 E:PK Q:100

```
$OFFTEXT  
$SYSINCLUDE mpsgeset M22
```

```
PW.FX = 1;
```

```
$INCLUDE M22.GEN  
SOLVE M22 USING MCP;
```

```
*           Counterfactual: 100% tax on X sector inputs:
```

```
TX = 1.0;  
$INCLUDE M22.GEN  
SOLVE M22 USING MCP;
```

```
* Algebraic representation -- note the complexity of two-  
* level CES functions which are automatically generated  
* within MPSGE.
```

EQUATIONS

PRF_X Zero profit for sector X
 PRF_Y Zero profit for sector Y
 PRF_W Zero profit for sector W (welfare index)

MKT_X Supply-demand balance for commodity X
 MKT_Y Supply-demand balance for commodity Y
 MKT_L Supply-demand balance for primary factor L
 MKT_K Supply-demand balance for primary factor L
 MKT_W Supply-demand balance for aggregate demand

I_CONS Income definition for CONS;

$$\text{PRF_X.. } 120 * \left(\frac{1}{6} * \text{PY}^{(1-0.5)} + \frac{5}{6} * (\text{PL}^{0.4} * \text{PK}^{0.6} * (1+\text{TX}))^{(1-0.5)} \right) ** \left(\frac{1}{(1-0.5)} \right) = \text{E} = 120 * \text{PX};$$

$$\text{PRF_Y.. } 120 * \left(\frac{1}{6} * \text{PX}^{(1-0.75)} + \frac{5}{6} * (\text{PL}^{0.6} * \text{PK}^{0.4})^{(1-0.75)} \right) ** \left(\frac{1}{(1-0.75)} \right) = \text{E} = 120 * \text{PY};$$

$$\text{PRF_W.. } 200 * \text{PX}^{0.5} * \text{PY}^{0.5} = \text{E} = 200 * \text{PW};$$

$$\text{MKT_X}.. 120 * X =E= 100 * W * \text{PX}^{**0.5} * \text{PY}^{**0.5} / \text{PX} + 20 * Y * (\text{PY} / \text{PX})^{**0.75};$$

$$\text{MKT_Y}.. 120 * Y =E= 100 * W * \text{PX}^{**0.5} * \text{PY}^{**0.5} / \text{PY} + 20 * X * (\text{PX} / \text{PY})^{**0.5};$$

$$\text{MKT_W}.. 200 * W =E= \text{CONS} / \text{PW};$$

$$\begin{aligned} \text{MKT_L}.. 100 =E= & 40 * X * \\ & (\text{PX} / ((1 + \text{TX}) * \text{PL}^{**0.4} * \text{PK}^{**0.6}))^{**0.5} \\ & * \text{PL}^{**0.4} * \text{PK}^{**0.6} / \text{PL} + \\ & 60 * Y * (\text{PY} / (\text{PL}^{**0.6} * \text{PK}^{**0.4}))^{**0.75} \\ & * \text{PL}^{**0.6} * \text{PK}^{**0.4} / \text{PL}; \end{aligned}$$

$$\begin{aligned} \text{MKT_K}.. 100 =E= & 60 * X * \text{PX} / ((1 + \text{TX}) * \text{PL}^{**0.4} * \text{PK}^{**0.6})^{**0.5} \\ & * \text{PL}^{**0.4} * \text{PK}^{**0.6} / \text{PK} + \\ & 40 * Y * (\text{PY} / (\text{PL}^{**0.6} * \text{PK}^{**0.4}))^{**0.75} \\ & * \text{PL}^{**0.6} * \text{PK}^{**0.4} / \text{PK}; \end{aligned}$$

$$\begin{aligned} \text{I_CONS}.. \text{CONS} =E= & 100 * \text{PL} + 100 * \text{PK} + \\ & \text{TX} * 100 * X * \text{PL}^{**0.4} * \text{PK}^{**0.6} * \\ & ((1 + \text{TX}) * \text{PL}^{**0.4} * \text{PK}^{**0.6})^{**0.5}; \end{aligned}$$


```
MODEL ALGEBRAIC /PRF_X.X, PRF_Y.Y, PRF_W.W, MKT_X.PX,  
MKT_Y.PY, MKT_L.PL,  
MKT_K.PK, MKT_W.PW, I_CONS.CONS /;
```

* Check the benchmark:

```
X.L=1; Y.L=1; W.L=1; PX.L=1; PY.L=1; PK.L=1; PW.L=1;  
CONS.L=200;
```

```
TX = 0;  
SOLVE ALGEBRAIC USING MCP;
```

* Solve the same counterfactual:

```
TX = 1;  
SOLVE ALGEBRAIC USING MCP;
```

Exercices:

- (1) Revise the X sector production block so that the elasticity of substitution between Y and value added is 0: there are fixed coefficients between Y and value added.

```

$PROD:X  s:0  va:1
          O:PX  Q:120
          I:PY  Q: 20
          I:PL  Q:40  va:      A:CONS  T:TX
          I:PK  Q:60  va:      A:CONS  T:TX

```

run the model and verify that the benchmark is reproduced. See if you can interpret the differences between the new counterfactual results and the initial results.

- (2) Rewrite the algebraic model in accordance with the new value of the upper level nest, and verify that you obtain identical solution values.

Model 23 introduces joint production

This model continues the same theme, by noting that a given sector can have several outputs as well as inputs.

Here we assume that two sectors, A and B, both produce both final outputs X and Y, but that sector A is relatively specialized in producing good X and sector B more specialized in producing good Y.

Here is the production block for B. The transformation elasticity between the two goods is equal to 1.5 (t:1.5) and the substitution elasticity between inputs is 1 (s:1).

```
$PROD:B t:1.5 s:1
      O:PX      Q:20
      O:PY      Q:80
      I:PL      Q:60
      I:PK      Q:40
```

Now the model itself, followed by the MCP version.

\$TITLE Model M23: Closed Economy 2x2 with Joint Production

\$ONTEXT

Markets	Production Sectors			Consumers
	A	B	W	CONS
PX	80	20	-100	
PY	20	80	-100	
PW			200	-200
PL	-40	-60		100
PK	-60	-40		100

\$OFFTEXT

PARAMETERS

TA;

TA = 0;

\$ONTEXT

\$MODEL: M23

\$SECTORS:

A ! Activity level for sector A (80:20 for X:Y)
B ! Activity level for sector B (20:80 for X:Y)
W ! Activity level for sector W (Hicksian welfare
index)

\$COMMODITIES:

PX ! Price index for commodity X
PY ! Price index for commodity Y
PL ! Price index for primary factor L
PK ! Price index for primary factor K
PW ! Price index for welfare (expenditure function)

\$CONSUMERS:

CONS ! Income level for consumer CONS

\$PROD:A t:2.0 s:1
O:PX Q:80
O:PY Q:20
I:PL Q:40.0 A:CONS T:TA
I:PK Q:60.0 A:CONS T:TA

\$PROD:B t:1.5 s:1
O:PX Q:20
O:PY Q:80
I:PL Q:60
I:PK Q:40

\$PROD:W s:1
O:PW Q:200
I:PX Q:100
I:PY Q:100

\$DEMAND:CONS
D:PW Q:200
E:PL Q:100
E:PK Q:100

```
$OFFTEXT  
$SYSINCLUDE mpsgeset M23
```

```
PW.FX = 1;
```

```
$INCLUDE M23.GEN  
SOLVE M23 USING MCP;
```

```
*           Counterfactual: 10% tax on X sector inputs:
```

```
TA = 0.10;  
$INCLUDE M23.GEN  
SOLVE M23 USING MCP;
```

```
*           Counterfactual: 100% tax on X sector inputs:
```

```
TA = 1.00;  
$INCLUDE M23.GEN  
SOLVE M23 USING MCP;
```

* now the mcp version, which again shows you the simplifying
*features of MPS/GE

EQUATIONS

PRF_A Zero profit for sector X
 PRF_B Zero profit for sector Y
 PRF_W Zero profit for sector W (welfare index)

MKT_X Supply-demand balance for commodity X
 MKT_Y Supply-demand balance for commodity Y
 MKT_L Supply-demand balance for primary factor L
 MKT_K Supply-demand balance for primary factor
 MKT_W Supply-demand balance for aggregate demand

I_CONS Income definition for CONS;

* Write the profit constraints as inequalities -- the tax
 * can cause sector A to shut down completely:

$$\begin{aligned} \text{PRF_A} \dots & 100 * \text{PL}^{**0.4} * \text{PK}^{**0.6} * (1+\text{TA}) = \text{G} = \\ & 100 * (0.8 * \text{PX}^{** (1+2.0)} + 0.2 * \\ & \text{PY}^{** (1+2.0)})^{** (1 / (1+2.0))}; \end{aligned}$$

$$\begin{aligned} \text{PRF_B}.. & 100 * \text{PL}^{**0.6} * \text{PK}^{**0.4} =\text{G}= \\ & 100 * (0.2 * \text{PX}^{** (1+1.5)} + 0.8 * \\ & \text{PY}^{** (1+1.5)})^{** (1 / (1+1.5))}; \end{aligned}$$

$$\text{PRF_W}.. 200 * \text{PX}^{**0.5} * \text{PY}^{**0.5} =\text{E}= 200 * \text{PW};$$

$$\begin{aligned} \text{MKT_X}.. & 80 * \text{A} * \\ & (\text{PX} / (0.8 * \text{PX}^{** (1+2.0)} + 0.2 * \text{PY}^{** (1+2.0)}))^{** (1 / (1+2.0))} ** 2 \\ & + 20 * \text{B} * \\ & (\text{PX} / (0.2 * \text{PX}^{** (1+1.5)} + 0.8 * \text{PY}^{** (1+1.5)}))^{** (1 / (1+1.5))} ** 1.5 \\ & =\text{E}= 100 * \text{W} * \text{PX}^{**0.5} * \text{PY}^{**0.5} / \text{PX}; \end{aligned}$$

$$\begin{aligned} \text{MKT_Y}.. & 20 * \text{A} * \\ & (\text{PY} / (0.8 * \text{PX}^{** (1+2.0)} + 0.2 * \text{PY}^{** (1+2.0)}))^{** (1 / (1+2.0))} ** 2.0 \\ & + 80 * \text{B} * \\ & (\text{PY} / (0.2 * \text{PX}^{** (1+1.5)} + 0.8 * \text{PY}^{** (1+1.5)}))^{** (1 / (1+1.5))} ** 1.5 \\ & =\text{E}= 100 * \text{W} * \text{PX}^{**0.5} * \text{PY}^{**0.5} / \text{PY}; \end{aligned}$$

$$\text{MKT_W}.. 200 * \text{W} =\text{E}= \text{CONS} / \text{PW};$$

$$\begin{aligned} \text{MKT_L}.. & 100 =\text{E}= 40 * \text{A} * \text{PL}^{**0.4} * \text{PK}^{**0.6} / \text{PL} + \\ & 60 * \text{B} * \text{PL}^{**0.6} * \text{PK}^{**0.4} / \text{PL}; \end{aligned}$$

```
MKT_K.. 100 =E= 60 * A * PL**0.4 * PK**0.6 / PK +
              40 * B * PL**0.6 * PK**0.4 / PK;
```

```
I_CONS.. CONS =E= 100*PL + 100*PK +
TA*100*A*PL**0.4*PK**0.6;
```

```
MODEL ALGEBRAIC /PRF_A.A, PRF_B.B, PRF_W.W, MKT_X.PX,
                MKT_Y.PY, MKT_L.PL,
                MKT_K.PK, MKT_W.PW, I_CONS.CONS /;
```

* Check the benchmark:

```
A.L=1; B.L=1; W.L=1; PX.L=1; PY.L=1; PK.L=1; PW.L=1;
CONS.L=200;
```

```
TA = 0;
SOLVE ALGEBRAIC USING MCP;
```

* Solve the same counterfactuals:

```
TA = 0.10;
SOLVE ALGEBRAIC USING MCP;
```

TA = 1.00;
 SOLVE ALGEBRAIC USING MCP;

Exercises:

- (1) Try a higher elasticity of transformation between output in the two sectors, such as $t = 10$. Try to guess ahead of time as to how this might affect the equilibrium outputs of the two sectors following imposition of the tax. Recall that the A sector is the sector which is relatively good at producing X.
- (3) Assume that the elasticity of transformation between outputs in each sector is zero: that is, outputs come in fixed proportions. Run the mpsge model. Now see if you can modify the algebraic model with this new assumption.
- (2) While it is harder to find a good discussion of CET functions in a textbook (which generally concentrate on the CES input side), see if you can verify that the MCP version is correct for one of the sectors. Let σ be the elasticity of transformation. CET functions have the form:

$$(aX^\alpha + bY^\alpha)^{1/\alpha} = F(L, K) \quad \text{where } \alpha \geq 1 \quad \sigma = \frac{1}{\alpha - 1}$$

Model 24 introduces specific factors

There are some inherent difficulties in Heckscher-Ohlin type models, in which all factors are used in and mobile between sectors.

In particular, trade theory demonstrates that when there are more goods than factors, there are “flats” in the aggregate transformation surface of the economy, and open economies tend to specialize in only as many goods as factors.

There are several ways around this problem. One is the “Armington” assumption introduced in the next chapter.

A second is to assume a portion of capital, or other factors (e.g., resources, land) is fixed or sector specific in each sector. That is what we do in this example.

We take the data matrix of M21 (M1_MPS) split capital into three factors, capital that is mobile between the two sectors, capital specific to Y and capital specific to X.

Markets	Production Sectors			Consumers
	X	Y	W	CONS
PX	100		-100	
PY		100	-100	
PW			200	-200
PL	-25	-75		100
PK	-75	-25		100

Now designate part of the capital in each sector as fixed, creating a four-factor model.

Markets	Production Sectors			Consumers
	X	Y	W	CONS
PX	100		-100	
PY		100	-100	
PW			200	-200
PL	-25	-75		100
PK	-50	-15		65
PKX	-25			25
PKY		-10		10

\$TITLE Model M24: Closed Economy 2x2 with Specific Factors

\$ONTEXT

Here is the initial data matrix for example M21 (also M1_MPS). As noted in the text description, it is technically useful to interpret a portion of capital in each sector as sector specific. Or it can in fact be a separate factor such as land or resources.

Markets	Production Sectors			Consumers
	X	Y	W	CONS
PX	100		-100	
PY		100	-100	
PW			200	-200
PL	-25	-75		100
PK	-50	-15		65
PKX	-25			25
PKY		-10		10

\$OFFTEXT

PARAMETERS

TX;

TX = 0;

\$ONTEXT

\$MODEL:M24

\$SECTORS:

X ! Activity level for sector X
Y ! Activity level for sector Y
W ! Activity level for sector W (welfare index)

\$COMMODITIES:

PW ! Price index for welfare (expenditure function)
PX ! Price index for commodity X
PY ! Price index for commodity Y
PL ! Price index for primary factor L
PK ! Price index for (mobile) capital
PKX ! Price index for sector-specific input for X
PKY ! Price index for sector-specific input Y

\$CONSUMERS:

CONS ! Income level for consumer CONS

\$PROD:X s:1
O:PX Q:100
I:PL Q: 25 A:CONS T:TX
I:PK Q: 50 A:CONS T:TX
I:PKX Q: 25 A:CONS T:TX

\$PROD:Y s:1
O:PY Q:100
I:PL Q: 75
I:PK Q: 15
I:PKY Q: 10

\$PROD:W s:1
O:PW Q:200
I:PX Q:100
I:PY Q:100

\$DEMAND:CONS

D:PW Q:200

E:PL Q:100

E:PK Q:65

E:PKX Q:25

E:PKY Q:10

\$OFFTEXT

\$SYSINCLUDE mpsgeset M24

\$INCLUDE M24.GEN

SOLVE M24 USING MCP;

* Solve a counterfactual:

TX = 0.5;

\$INCLUDE M24.GEN

SOLVE M24 USING MCP;

Exercise:

As an exercise after examining this model, try to guess how the introduction of the specific factors affects the responsiveness (elasticity) of X and Y outputs to the 100% tax. Run the model and compare it to the 50% X-sector tax results in model M21 (M1-MPS). See if you can interpret the results.

Model 25 introduces and initially slack activity

As noted in chapter one, an attractive and powerful feature of MPS/GE is that it solves complementarity problems in which some production activities can be slack for some values of parameters and active for others.

This allows researchers to consider a much wider set of problems that is allowed using software which can only solve systems of equations.

Model 25 presents a simple example, motivated by tax evasion activities. There is a third sector, Z, which also produces good X but it is 10% less efficient (10% more costly) than the X activity itself. So initially, Z does not operate.

But when a tax of 25% is imposed on X, this activity goes slack and Z begins to operate.

We could think of Z as a tax evasion or “informal” activity that is less efficient but can successfully avoid the tax.

\$TITLE Model M25: Closed 2x2 Economy with an Unprofitable
*Activity

\$ONTEXT

Markets	Production Sectors			Consumers
	X	Y	W	CONS
PX	100		-100	
PY		100	-100	
PW			200	-200
PL	-40	-60		100
PK	-60	-40		100

Activity Z is unprofitable at initial equilibrium prices. It is therefore not operated, and we cannot infer its technical properties from the benchmark social accounting data. We assume that Z is 10% less efficient than X.

\$OFFTEXT

PARAMETERS

TX;

TX = 0;

\$ONTEXT

\$MODEL:M25

\$SECTORS:

X !Activity level for sector X
Y ! Activity level for sector Y
W ! Activity level for sector W (welfare index)
Z ! Alternative activity for producing X.

\$COMMODITIES:

PX ! Price index for commodity X
PY ! Price index for commodity Y
PL ! Price index for primary factor L
PK ! Price index for primary factor K
PW ! Price index for welfare

\$CONSUMERS:

CONS ! Income level for consumer CONS

\$PROD:X s:1

O:PX Q:100

I:PL Q: 40 A:CONS T:TX

I:PK Q: 60 A:CONS T:TX

\$PROD:Y s:1

O:PY Q:100

I:PL Q:60

I:PK Q:40

\$PROD:Z s:1

O:PX Q: 100

I:PL Q: 44

I:PK Q: 66

```
$PROD:W  s:1  
         O:PW      Q:200  
         I:PX      Q:100  
         I:PY      Q:100
```

```
$DEMAND:CONS  
         D:PW      Q:200  
         E:PL      Q:100  
         E:PK      Q:100
```

```
$OFFTEXT
```

```
$SYSINCLUDE mpsgeset M25
```

```
PW.FX = 1;
```

```
Z.L = 0;
```

```
$INCLUDE M25.GEN
```

```
SOLVE M25 USINCP MCP;
```

* Lets levy a high on sector X and see what happens:


```
TX = 0.25;  
$INCLUDE M25.GEN  
SOLVE M25 USING MCP;
```

* What is the effect of the tax if Z could not be used?

```
Z.FX = 0;  
TX = 0.25;  
  
$INCLUDE M25.GEN  
SOLVE M25 USING MCP;
```

Exercise:

As a second counterfactual, we impose the tax but do not let the Z sector operate by imposing the restriction $Z.FX = 0$;

(1) Compare the results of this run to the first counterfactual in which Z is allowed to operate. Can you interpret the welfare results? Hint: while the tax is distortionary, the switch to the inefficient activity uses real resources to avoid the tax.

(2) Now raise the tax to 100%. Does this result that tax evasion (the switch to Z) is welfare worsening still hold?

Model M26 Introduces a labor supply or labor/leisure choice activity

Often general-equilibrium models used in international trade assume that factors of production, especially labor, are in fixed and inelastic supply.

But designing tax, welfare, and education systems, endogenizing labor supply is a crucial part of the story.

Model M26 endogenizes labor supply, allowing labor to chose between leisure and labor supply with leisure entering into the workers utility function.

In our formulation, we introduce an additional activity T, which transforms leisure (price PL) into labor supplied (price PLS).

\$PROD:T

O:PLS Q:100

I:PL Q:100 A:CONS T:TL

The production block for W specifies a nesting structure in which goods have an elasticity of substitution of 1 between them in a lower nest, and goods and leisure have an elasticity of substitution between them of 0.5.

```
$PROD:W  s:0.5  cons:1  
O:PW    Q:300  
I:PX    Q:100  cons:  
I:PY    Q:100  cons:  
I:PL    Q:100
```

The consumer is assumed to be endowed with 200 units of labor/leisure (found in the DEMAND block), of which 100 units are supplied to the labor market initially.

\$TITLE Model M26: 2x2 Economy with Labor-Leisure Choice

\$ONTEXT

Activity T transforms leisure into labor supply:

Markets	Production Sectors				Consumers
	A	B	W	T	CONS
PX	80	20	-100		
PY	20	80	-100		
PW			300		-300
PLS	-40	-60		100	
PL			-100	-100	200
PK	-60	-40			100

\$OFFTEXT

PARAMETERS

TL

WELFARE

REALCONS;

TL = 0;

\$ONTEXT

\$MODEL:M26

\$SECTORS:

X	!	Activity level for sector X
Y	!	Activity level for sector Y
T	!	Labor supply
W	!	Activity level for sector W (welfare index)

\$COMMODITIES:

PX	!	Price index for commodity X
PY	!	Price index for commodity Y
PL	!	Price index for leisure
PLS	!	Price index for labor supply (factor L input)
PK	!	Price index for primary factor K

PW ! Price index for welfare (expenditure function)

\$CONSUMERS:

CONS ! Income level for consumer CONS

\$PROD:X s:1

O:PX Q:100

I:PLS Q:40

I:PK Q:60

\$PROD:Y s:1

O:PY Q:100

I:PLS Q:60

I:PK Q:40

\$PROD:T

O:PLS Q:100

I:PL Q:100 A:CONS T:TL

```
$PROD:W  s:0.5  cons:1
          O:PW   Q:300
          I:PX   Q:100  cons:
          I:PY   Q:100  cons:
          I:PL   Q:100
```

```
$DEMAND:CONS
          D:PW   Q:300
          E:PL   Q:200
          E:PK   Q:100
```

```
$OFFTEXT
```

```
$SYSINCLUDE mpsgeset M26
```

```
PW.FX = 1;
```

```
$INCLUDE M26.GEN
```

```
SOLVE M26 USING MCP;
```

```
WELFARE = W.L;
```



```
REALCONS = (PX.L*X.L*100 +
PY.L*Y.L*100) / (PX.L**0.5*PY.L**0.5*200);
```

```
DISPLAY WELFARE, REALCONS;
```

* Solve a counter-factual, tax labor supply at 25%

```
TL = 0.5;
```

```
$INCLUDE M26.GEN
```

```
SOLVE M26 USING MCP;
```

```
WELFARE = W.L;
```

```
REALCONS = (PX.L*X.L*100 +
PY.L*Y.L*100) / (PX.L**0.5*PY.L**0.5*200);
```

```
DISPLAY WELFARE, REALCONS;
```

Several features of this program require some explanation. First, the use of the labor supply activity is going to imply two separate prices in the presence of the labor tax.

There is PL, which is the price of leisure, or alternatively the consumer's price of a labor supplied to the market. PLS is going to be the producer's cost of labor. The two are related by

$$PLS = PL(1+TL)$$

We could also refer to these as the before (PL) and after tax wage (PLS). Second, note the equations and notation at the end of the program.

```
WELFARE = W.L;
REALCONS = (PX.L*X.L*100 + PY.L*Y.L*100)
            / (PX.L**0.5*PY.L**0.5*200);

DISPLAY WELFARE, REALCONS;
```

A “.L” after a variable asks for the current value of a variable. Thus “W.L” gives the value of the variable named “W”.

In the first two of these statements, parameters (declared earlier) are assigned values following the solution to the model. The first is just the value of welfare.

The second, “REALCONS” is short for the real value of goods consumption.

As you will see if you run this model, the labor tax leads to a reduction in labor supply.

This of course leads to a fall in commodity consumption but also to a rise in leisure.

Thus in this case, the usual statistics overstate the burden of the tax and would overstate the benefit of removing the tax if labor supply increases.

Finally, GAMS does not automatically write out the values of parameters in the listing file (in this case M26.LST). You have to request that, which is done here with the `DISPLAY` statement as shown. Parameter values will be written out following the values of the variables of the model.

Exercise:

Change the elasticity of substitution between leisure and goods and see how this affects the results concerning the labor tax.

Model M27

This is a model which may be of interest to development and public finance economists. It assumes that there are two labor markets, a “formal” and an “informal” market.

Governments are able to collect taxes on the former but not on the latter. The representative household can choose how much labor to supply to each market.

For simplicity, we assume that there is no labor-leisure decision, and that all labor is supplied to one of the two markets, but that can be very easily added and indeed we will suggest that as an exercise at the end.

There are many ways of doing this. We first of all use an activity denoted LS which takes household labor and produces two outputs, formal and informal labor (prices PLSF and PLSI) according to a CES transformation function with an elasticity 5.0.

Think of this as a household technology embodying the fact that the two types of labor are not perfect substitutes in supply. For example, this might be a crude simplification of the fact that the representative household is actually many

households (or household locations) some of whom are better at supplying formal labor and vice versa. Concentrating supply in either market leads to something like “diminishing returns”, a concave transformation frontier between the two types of labor.

In addition, the two types of labor can be imperfect substitutes on the production side. That is what we assume here. Only formal labor is used in the X sector, while both formal and (mostly) informal labor are used in the Y sector. The two types of labor are in a lower level nest with an elasticity of substitution of 3.

The formal and informal labor supplies could go directly into production, but the listing file will not directly tell us how much of each type is produced by activity LS.

Thus we use two “dummy” activities LF and LI which take a unit of formal labor (LF) or informal labor (LI) and just turn each unit into a unit of the same thing with a different commodity name (PLF and PLI) which are the actual inputs into production.

This is just a trick so that the listing file will tell us how much of each type of labor is supplied, the activity levels of LF and LI respectively. In addition, this trick

is convenient in multi-sector models because the tax on formal labor need only be specified once, in the LF activity, and not in every production block using formal labor.

Here is the program. You will see from the listing file that the tax on formal labor supply leads to a large shift of household supply toward informal labor and that there is a large shift in output toward Y, the sector using informal labor.

\$TITLE Model M27: 2x2 Economy with Formal/Informal Labor Supply

\$ONTEXT

Activity LS transforms leisure into formal and informal labor supplies. LF and LI are "dummy" activities used to keep track of how much labor is supplied to each market.

Markets	Production Sectors						Consumers
	X	Y	W	LF	LI	LS	CONS
PX	100		-100				
PY		100	-100				
PW			200				-200
PLF	-40	-10		50			
PLI		-50			50		
PLSF			-50			50	
PLSI					-50	50	
PL						-100	100
PK	-60	-40					100

\$OFFTEXT

PARAMETERS

TL;

TL = 0;

\$ONTEXT

\$MODEL:M27

\$SECTORS:

X ! Activity level for sector X

Y ! Activity level for sector Y

LS ! Activity level for household labor supply

LF ! Activity for formal labor supply

LI ! Activity for informal labor supply

W ! Activity level for sector W (Hicksian

welfare index)

\$COMMODITIES:

PX ! Price index for commodity X
 PY ! Price index for commodity Y
 PL ! Price index for labor
 PLSF ! Price index for formal labor supplied to market
 PLSI ! Price index for informal labor supply to market
 PLF ! Price index for formal labor supplied to firms
 PLI ! Price index for informal labor supplied to firms
 PK ! Price index for primary factor K
 PW ! Price index for welfare (expenditure function)

\$CONSUMERS:

CONS ! Income level for consumer CONS

\$PROD:X s:1

O:PX Q:100
 I:PLF Q: 40
 I:PK Q: 60

\$PROD:Y s:1 a:3
O:PY Q:100
I:PLF Q: 10 a:
I:PLI Q: 50 a:
I:PK Q: 40

\$PROD:LS t:5.0
O:PLSF Q: 50
O:PLSI Q: 50
I:PL Q:100

\$PROD:LF
O:PLF Q: 50
I:PLSF Q: 50 A:CONS T:TL

\$PROD:LI
O:PLI Q: 50
I:PLSI Q: 50

```
$PROD:W s:1.0
```

```
O:PW Q:200
```

```
I:PX Q:100
```

```
I:PY Q:100
```

```
$DEMAND:CONS
```

```
D:PW Q:200
```

```
E:PL Q:100
```

```
E:PK Q:100
```

```
$OFFTEXT
```

```
$SYSINCLUDE mpsgeset M27
```

```
PW.FX = 1;
```

```
$INCLUDE M27.GEN
```

```
SOLVE M27 USING MCP;
```

```
* Solve a counter-factual, tax formal labor supply at 50%
```

```
TL = 0.5;
```

```
$INCLUDE M27.GEN
```

```
SOLVE M27 USING MCP;
```

Model M28

A great many questions of interest to trade and public finance economists involve issues of distribution rather than or in addition to issues of aggregate welfare. Households (or “consumers” in MPS/GE) may differ in their preferences and more importantly in their sources of income (or their factor endowments).

Adding multiple household types is a straightforward extension of our earlier models. In model M28, we allow for two households. Household A is relatively well endowed with labor, and also has a preference for good Y, which is the labor-intensive good. Household B is relatively well endowed with capital and has a relative preference for the capital intensive good.

Our counterfactual experiment is to place a tax on the factor inputs to X, assigning half the revenue to each consumer. As you will guess, this tax lowers the welfare of household B. However, the redistribution effect outweighs the overall deadweight loss of the tax for household A, which is actually better off. This welfare gain is a combination of a redistribution in favor of capital, and a lowering of the relative consumer price of Y, the good favored by household A.

\$TITLE Model M28.GMS: 2x2 Economy with Two Household Types

\$ONTEXT

Two household: differ in preferences and in endowments

Markets	Production Sectors				Consumers	
	X	Y	WA	WB	A	B
PX	100		-40	-60		
PY		100	-60	-40		
PWA			100		-100	
PWB				100		-100
PL	-25	-75			90	10
PK	-75	-25			10	90

\$OFFTEXT

PARAMETERS

TX;

TX = 0;

\$ONTEXT

\$MODEL:M28

\$SECTORS:

X	!	Activity level for sector X
Y	!	Activity level for sector Y
WA	!	Welfare index for consumer A
WB	!	Welfare index for consumer B

\$COMMODITIES:

PX	!	Price index for commodity X
PY	!	Price index for commodity Y
PL	!	Price index for primary factor L
PK	!	Price index for primary factor K
PWA	!	Price index for consumer A welfare
PWB	!	Price index for consumer B welfare

\$CONSUMERS:

CONSA ! Income level for consumer A
 CONSB ! Income level for consumer B

- * More than one tax may be levied on a single transaction
- * with
- * revenues accruing to different agents. As specified, the
- * ad-valorem tax rate on inputs to sector X equals 2*TX.
- * Half of the tax revenue accrues to A and half to B.

\$PROD:X s:1

O:PX Q:100
 I:PL Q: 25 A:CONSA T:TX A:CONSB T:TX
 I:PK Q: 75 A:CONSA T:TX A:CONSB T:TX

\$PROD:Y s:1

O:PY Q:100
 I:PL Q: 75
 I:PK Q: 25

\$PROD:WA s:1

O:PWA Q:100

I:PX Q:40

I:PY Q:60

\$PROD:WB s:1

O:PWB Q:100

I:PX Q:60

I:PY Q:40

\$DEMAND:CONSA

D:PWA Q:100

E:PL Q: 90

E:PK Q: 10

\$DEMAND:CONSB

D:PWB Q:100

E:PL Q: 10

E:PK Q: 90

\$OFFTEXT


```
$SYSINCLUDE mpsgeset M28
```

```
PL.FX = 1;
```

```
$INCLUDE M28.GEN  
SOLVE M28 USING MCP;
```

```
*           Solve a counterfactual:
```

```
TX = 0.25;  
$INCLUDE M28.GEN  
SOLVE M28 USING MCP;
```

Exercises:

- (1) Recalibrate the data so that the households have the same preferences. Running the experiment gives the a welfare effect due only to the change in factor prices following the imposition of the tax.

- (2) You might (correctly) guess that there is no way to redistribute the tax unevenly and make both households worse off. That would violate the first theorem of welfare economics. Try some alternative distributions to check on this. You will need to specify two different tax parameters, but they should continue to sum to $0.5 = 2 * 0.25$.

Model 29

No budget study we are aware of has ever suggested that consumer preferences are homogeneous. Households tend to spend a much higher proportion of their income on food at low incomes, for example, than at high incomes.

One alternative formulation is known as the Stone-Geary utility function, which in turn gives rise to the linear expenditure (LES) system of demand equations, the latter being popular in budget studies. The Stone-Geary utility function is just a Cobb-Douglas function with the origin displaced from zero. These displacements, if positive, are typically called “minimum consumption requirements”, meaning that the consumer gets no positive utility until these needs are met.

$$U = (X - X^*)^\alpha Y^{1 - \alpha}$$

If we maximized this subject to the usual budget constraint with income I , the demand functions for X and Y would be:

$$X = X^* + \frac{\alpha(I - p_x X^*)}{p_x} \quad Y = \frac{(1 - \alpha)(I - p_x X^*)}{p_y} \quad p_x X + p_y Y = I$$

The first equation is rather intuitive in words. It says that you first purchase the minimum consumption requirements, and then you spend a constant fraction (α) of remaining income net of the minimum requirements on X. Further algebra would give us the budget share spent on X and the income elasticity of demand for X.

$$\frac{p_x X}{I} = \frac{\alpha I + (1 - \alpha)p_x X^*}{I} \quad \frac{I}{X} \frac{dX}{dI} = \frac{\alpha I}{\alpha I + (1 - \alpha)p_x X^*}$$

The budget share spent on X falls with increases in income, asymptotically approaching α as income rises. The income elasticity of demand rises with income, asymptotically approaching 1.

Suppose that we want to calibrate our initial data to the assumption that the

income elasticity of demand for X is initially equal to 0.75. If we solve the share equation (equal to 0.5) in the data and the income-elasticity equation (equal to 0.75 by assumption), we get $\alpha = 3/8$. This will then allow us to solve for X^* , which is $X^* = 40$. The trick is then to revise the benchmark data matrix, giving the consumer a negative endowment of $X = 40$. The utility (welfare) function W then has an input of 60 units of X (100 minus the minimum consumption requirement) and 100 units of Y . At prices of 1 for each good, MPS/GE will then calibrate the Cobb-Douglas utility function with an $\alpha = 3/6$ (60/160).

The counterfactual experiment in this model is to double the consumers endowment. Note from the results that there is a shift in consumption toward Y , the high income-elasticity good. Of course, the change in X and Y consumption cannot be directly interpreted as income elasticities of demand, since prices will change in general equilibrium. The price of Y will rise relative to X , and the price of the factor used intensively in Y will rise relative to the price of the other factor.

\$TITLE Model M29: Closed 2x2 Economy -- Stone Geary (LES)
 *Preferences

\$ONTEXT

The observed data is:

Markets	Production Sectors			Consumers
	X	Y	W	CONS
PX	100		-100	
PY		100	-100	
PW			200	-200
PL	-40	-60		100
PK	-60	-40		100

But calibrated to the model as:

Markets	Production Sectors			Consumers
	X	Y	W	CONS
PX	100		-60	-40
PY		100	-100	
PW			160	-160
PL	-40	-60		100
PK	-60	-40		100

\$OFFTEXT

PARAMETERS

ENDOW;

ENDOW = 1;

\$ONTEXT

\$MODEL:M29

\$SECTORS:

X ! Activity level for sector X
Y ! Activity level for sector Y
W ! Activity level for sector W (welfare index)

\$COMMODITIES:

PX ! Price index for commodity X
PY ! Price index for commodity Y
PL ! Price index for primary factor L
PK ! Price index for primary factor K
PW ! Price index for welfare

\$CONSUMERS:

CONS ! Income level for consumer CONS

\$PROD:X s:1
O:PX Q:100
I:PL Q: 40
I:PK Q: 60

\$PROD:Y s:1
O:PY Q:100
I:PL Q: 60
I:PK Q: 40

\$PROD:W s:1
O:PW Q:160
I:PX Q: 60
I:PY Q:100

\$DEMAND:CONS
D:PW Q:160
E:PL Q: (100*ENDOW)
E:PK Q: (100*ENDOW)
E:PX Q: -40

\$OFFTEXT

```
$SYSINCLUDE mpsgeset M29
```

```
PW.FX = 1;
```

```
$INCLUDE M29.GEN  
SOLVE M29 USING MCP;
```

```
*           Counterfactual: double the factor endowment.
```

```
ENDOW = 2;
```

```
$INCLUDE M29.GEN  
SOLVE M29 USING MCP;
```